Métodos Determinísticos de Investigação Operacional

Trabalho Prático III

8 de Janeiro de 2021

a85635 André Nunes
a85517 Duarte Oliveira
a89561 Gustavo Lourenço
a89501 Martim Almeida

Mestrado Integrado em Engenharia Informática
Universidade do Minho

Conteúdo

1 Introdução

No âmbito do terceiro trabalho prático da unidade curricular de Modelos Determinísticos de Investigação operacional foi nos pedido que resolvêssemos um problema usando o **método do caminho crítico**. Para tal foi-nos apresentado um conjunto de várias atividades e as suas respetivas durações e precedências.
Este projeto será dividido em duas partes:

- A Parte 0, onde temos como o objetivo calcular a duração mínima para realizar todas as atividades respeitando as suas respectivas durações e restrições de precedência.
- A Parte 1, onde poderemos aumentar os recursos aplicados às atividades através de custos suplementares, podemos assim reduzir a duração total das atividades, estas reduções dependem das características de cada atividade. O objetivo desta parte será minimizar o custo para reduzir a duração consequente da realização de todas as atividades.

Em primeira instância podemos observar a tabela correspondente às diferentes atividades e suas respetivas durações e precedências.

Actividade	Duração	Precedências
0	4	–
1	6	0
2	7	1,4
3	2	2,5
4	9	0,7
5	4	4,8
6	5	–
7	6	6
8	4	7,10
9	2	8,11
10	8	6
11	7	10

Figura 1: Actividades e relações de precedência

Também nos foi fornecido o grafo inicial que corresponde à temática apresentada sob o qual nos basearemos para poder chegar a solução deste problema.



Figura 2: Grafo inicial associado a este projeto

2 Parte 0

Devido à restrição apresentada no enunciado, sendo o maior dos nossos números de aluno **89561**, na sua equivalência a $ABCDE$ equivale a adaptar o grafo, removendo a atividade 6 e 1, em seguida apresentado:



Figura 3: Grafo personalizado

Começando aqui a elaborar a resolução deste problema é importante apontar que existem muitas formas de determinar o caminho crítico, entre as quais se destacam o **caminho mais longo** e a **minimização do tempo de conclusão**. Por nossa decisão decidimos aplicar ambos e apresentaremos a formulação dos mesmos ao longo deste relatório.

2.1 Caminho mais longo

Sabe-se que entre o vértice que representa o inicio do projeto e o vértice que representa o fim do mesmo existe o **caminho crítico -** correspondente ao caminho mais longo entre os vértices referidos. Portanto, as atividades que pertençam a este caminho, serão também denominadas de **críticas**, visto que um atraso irá obrigatoriamente levar a um aumento no tempo de execução de todo o projeto. Isto significa que estas atividades serão aquelas a que têm que ser mais monitorizadas e controladas.

- a variável x_{ij} está associada a cada arco (i,j) do grafo, toma o valor **1** se o arco fizer parte do caminho mais longo e 0 caso contrário.
- a variável d_i representa a **duração da atividade i**.
- As restrições do problema traduzem a conservação de fluxo em cada vértice, ou seja:

  O número de unidades de fluxo que entram no vértice j através dos arcos (i,j) têm que ser igual ao número de unidades que dele saem através dos arcos (j,k).
  Se há um arco do caminho a entrar num vértice, deverá haver um arco a sair, ou inverso também se verifica - se não houver arco a entrar num vértice, o caminho não passará pelo vértice.
  Esta relação é válida para todos os vértices, excepção feita ao vértice *"ini"*, onde é introduzida uma unidade de fluxo na rede, e ao vértice *"fim"*, onde a unidade é removida. De facto, a restrição relativa ao vértice *"fim"* pode ser retirada do modelo, por ser linearmente dependente das outras restrições.
- a função objetivo, é calculada através de:

  $$\sum d_i \times x_{ij}$$
  $$\text{c/ } i \in \{0, 2, 3, 4, 5, 7, 8, 9, 10, 11\} \quad \text{e} \quad j \in \{2, 3, 4, 5, 8, 9, 11, fim\}$$

  É de notar que: i não pode tomar o valor "*fim*"pois não existe nenhum arco a sair deste vértice tal como j não pode tomar o valor "*ini*"pois não há nenhum arco a entrar neste vértice. Os arcos (ini,0) , (ini,7) e (ini,10) foram omitidos da função objetivo visto que o vértice *"ini"* tem duração nula. Por isso, i não toma o valor *"ini"* e j não toma os valores 0, 7 ou 10.

2.1.1 Ficheiro Input

É aqui discriminado o ficheiro de input, devidamente explicitado.

```
/* Função objetivo*/
max: 4 x02 + 4 x04 + 7 x23 + 2 x3fim + 9 x42 + 9 x45 + 4 x53+4 x5fim +
 6 x74 + 6 x78 + 4 x85 + 4 x89 + 2 x9fim + 8 x108 + 8 x1011 + 7 x119;

/*Restrições de precedência*/
vertice_i: xini0 + xini7 + xini10 = 1;
vertice_0: xini0 = x02 + x04;
vertice_2: x02 + x42 = x23;
vertice_3: x23 + x53 = x3fim;
vertice_4: x04 + x74 = x42 + x45;
vertice_5: x45 + x85 = x53 + x5fim;
vertice_7: xini7 = x74 + x78;
vertice_8: x78 + x108 = x85 + x89;
vertice_9: x89 + x119 = x9fim;
vertice_10: xini10 = x108 + x1011;
vertice_11: x1011 = x119;

Bin
xini0, xini7, xini10, x02, x04, x23, x3fim, x42, x45, x53,
x5fim, x74, x78, x85, x89, x9fim, x108, x1011, x119;
```

Figura 4: Ficheiro input

2.1.2 Ficheiro Output

Está aqui discriminado o ficheiro de Output do problema:

Variables	MILP ...	result
	24	24
x23	1	1
x3fim	1	1
x42	1	1
x74	1	1
xini7	1	1
x02	0	0
x04	0	0
x1011	0	0
x108	0	0
x119	0	0
x45	0	0
x53	0	0
x5fim	0	0
x78	0	0
x85	0	0
x89	0	0
x9fim	0	0
xini0	0	0
xini10	0	0

Figura 5: Ficheiro output

Através do *Ficheiro de Output* conseguimos definir quais variáveis de decisão x_{ij} que têm o valor 1, podendo assim **definir o Caminho Crítico do grafo**, que corresponde às atividades *7,4,2 e 3*, e assim calcular **a duração associada a este caminho**, que é 24.
Define-se sob contorno vermelho aquele que é o caminho crítico do nosso grafo.



Figura 6: Caminho Crítico do Grafo

2.2 Minimização do tempo de conclusão

Optamos também por definir este problema utilizando este modelo, de forma a poder também explora-lo, onde cada variável representa o tempo de inicio de uma determinada atividade, tendo como objetivo (como o nome indica) de **minimizar o tempo de execução total** deste problema. É de notar que para este problema é necessário fincar alguns pontos, tanto a nível matemático como interpretativo.

- a variável t_i representa tempo de início atividade i, com i $\in \{0, 2, 3, 4, 5, 7, 8, 9, 10, 11\}$.
- a variável d_i representa a duração de uma atividade i.
- a variável t_{ini} representa o tempo de ínicio do modelo.
- a variável t_{fim} representa o tempo em que se findou o modelo.
- o tempo até à conclusão de uma determinada atividade i é dado pela soma de $t_i + d_i$
- as restrições intrínsecas a este projeto traduzem **relações de precedência** entre atividades.

2.3 Ficheiro input

É aqui discriminado o ficheiro de input, devidamente explicitado.

```
/*Função Objetivo*/
min: tfim;

/*variáveis de decisão*/
t0 >= tini + 0; //A actividade ini precede a actividade 0
t2 >= t0 + 4; //A actividade 0 precede a actividade 2
t2 >= t4 + 9; //A actividade 4 precede a actividade 2
t3 >= t2 + 7; //A actividade 2 precede a actividade 3
t3 >= t5 + 4; //A actividade 5 precede a actividade 3
t4 >= t0 + 4; //A actividade 0 precede a actividade 4
t4 >= t7 + 6; //A actividade 7 precede a actividade 4
t5 >= t4 + 9; //A actividade 4 precede a actividade 5
t5 >= t8 + 4; //A actividade 8 precede a actividade 5
t7 >= tini + 0; //A actividade ini precede a actividade 7
t8 >= t7 + 6;  //A actividade 7 precede a actividade 8
t8 >= t10 + 8; //A actividade 10 precede a actividade 8
t9 >= t8 + 4; //A actividade 8 precede a actividade 9
t9 >= t11 + 7; //A actividade 11 precede a actividade 9
t10 >= tini + 0; //A actividade ini precede a actividade 10
t11 >= t10 + 8; //A actividade 10 precede a actividade 11
tfim >= t3 + 2; //A actividade 3 precede a actividade fim
tfim >= t5 + 4; //A actividade 5 precede a actividade fim
tfim >= t9 + 2; //A actividade 9 precede a actividade fim

/*restrições de integralidade*/
int tini, t0, t2, t3, t4, t5, t7, t8, t9, t10, t11, tfim;
```

Figura 7: Ficheiro input

2.4 Ficheiro output

Está aqui discriminado o ficheiro de Output do problema:

Variables	MILP ...	result
	24	24
tfim	24	24
t0	0	0
tini	0	0
t2	15	15
t4	6	6
t3	22	22
t5	18	18
t7	0	0
t8	11	11
t10	0	0
t9	15	15
t11	8	8

Figura 8: Ficheiro output

Através deste ficheiro de output conseguimos desenvolver o diagrama de Gantt associado este problema.



Figura 9: Diagrama de Gantt

Como podemos observar, através do Ficheiro de Output e através do Diagrama de Gantt, o tempo mínimo para este problema é 24.

2.5 Interpretação de resultados

De forma muito resumida, como era espectável, contando que a nossa formulação era a correta, o valor da duração do **caminho mais longo** e da **minimização do tempo de conclusão** eram idênticos - 24.

3 Parte 1

3.1 Análise e interpretação do problema

Este problema tem como objetivo concluir o problema da Parte 0 e reduzir o seu tempo de execução em 3 U.T..
Para que isso seja possível, nós teremos que considerar que é possível, ao aumentar os recursos aplicados e com custos suplementares, reduzir a duração de uma actividade. Teremos duas formas de reduzir a duração desta atividade:

- O aumento de custo é dado por uma função contínua côncava;
- O aumento de custo é ditado pela escolha de um conjunto discreto de opções.

Em relação ao primeiro caso, cada actividade tem cinco parâmetros adicionais: o primeiro é o valor do custo normal, expresso em unidades monetárias [U.M.], o segundo é o valor de c_1, o custo suplementar de reduzir a duração da actividade em uma unidade de tempo [U.T.], expresso em [U.M./U.T.], o terceiro é o valor da máxima redução de tempo a um custo c_1, o quarto é o valor de c_2, o custo suplementar de reduzir a duração da actividade em uma unidade de tempo [U.T.] após ter aplicado a máxima redução a um custo c_1, expresso em [U.M./U.T.], e o quinto é o valor da máxima redução de tempo a um custo c_2.
Estes valores estão apresentados na seguinte tabela:

Actividade	Custo Normal	c_1	Máx. red. a custo c_1	c_2	Máx. red. a custo c_2
0	400	200	0,5	100	0,5
1	1000	600	1	300	1
2	1400	1000	3	500	1
3	300	200	0,5	100	0,5
4	2000	800	2	400	1
5	1000	1600	0,5	800	0,5
6	800	180	1	90	1
7	900	–	–	–	–
8	600	200	0,5	100	0,5
9	300	–	–	–	–
10	1600	1000	0,5	500	0,5
11	1400	600	1	300	1

Figura 10: Tabela de Custos de Redução da duração das Atividades

É de notar que, como já foi referido anteriormente, no grafo que está a ser utilizado não constam as Atividades 1 e 6.

Em relação ao segundo caso, além das opções com a duração normal, as Atividades 7 e 9 podem ser realizadas com as seguintes opções alternativas:
A atividade 7 pode também ser realizada com uma duração de 5 unidades de tempo e com custo adicional de 300 U.M. ou com uma duração de 4 unidades de tempo e com um custo adicional de 1100 U.M.. A atividade 9 pode também ser realizada com uma duração de 1 unidade de tempo e com custo adicional de 200 U.M. ou com uma duração de 0 unidades de tempo e com um custo adicional de 400 U.M..

Podemos concluir que este é um problema de *crashing times* de gestão de projetos, isto é, é um problema onde há interesse de reduzir o tempo de execução do projeto e, embora haja uma duração definida para cada atividade, é muitas vezes possível, aumentando os recursos nela aplicados, reduzir a sua duração. Isto é feito com custos suplementares.

3.2 Solução do Problema

3.2.1 Aproximação a uma Função Linear por Partes

Neste problema, podemos resolver o aumento de custo por uma função contínua côncava através de um conjunto de funções lineares por partes.
Dado:

- Uma variável c_1, que representa o custo suplementar de reduzir a duração da atividade em uma unidade de tempo;
- Uma variável m_1, que representa a máxima redução de tempo de c_1;
- Uma variável c_2, que representa o custo suplementar de reduzir a duração da atividade em uma unidade de tempo após ter aplicado a máxima redução a um custo c_1;
- Uma variável m_2, que representa a máxima redução de tempo de c_2;
- Uma variável x, que representa o valor reduzido em unidades de tempo;

Podemos representar esta função côncava, a partir da seguinte função linear por partes:

$$f(x) = \begin{cases} c_1 \times x, & \text{se } 0 \leq x \leq m_1 \\ c_1 \times m_1 + c_2 \times (x - m_1), & \text{se } m_1 \leq x \leq m_1 + m_2 \end{cases}$$

Posteriormente, podemos simplificar este sistema, obtendo:

$$f(x) = \begin{cases} c_1 \times x, & \text{se } 0 \leq x \leq m_1 \\ (c_1 - c_2) \times m_1 + c_2 \times x, & \text{se } m_1 \leq x \leq m_1 + m_2 \end{cases}$$

Uma forma de modelar esta função é a seguinte:

$$\begin{gathered} f(x) = (0y_1 + c_1 \times x_1) + (((c_1 - c_2) \times m_1) \times y_2 + c_2 \times x_2) \\ 1 = y_1 + y_2 \\ x = x_1 + x_2 \\ 0y_1 \leq x_1 \leq m_1 y_1 \\ m_1 y_2 \leq x_2 \leq (m_1 + m_2) \times y_2 \\ x_i \geq 0 \text{ e inteiro, i} \in \{1, 2\} \\ y_i \in \{0, 1\}, \text{ i} \in \{1, 2\} \end{gathered}$$

Para cada uma das atividades do grafo, nós conseguimos obter a função objetivo **através do somatório de todas as funções por partes das atividades**. Porém, teremos que acrescentar as restrições relativas a esta modelação das funções.

3.2.2 Variáveis de decisão

Para a resolução deste problema foram utilizadas as seguintes variáveis de decisão:

- A variável t_i representa tempo de início atividade i, com i $\in \{0, 2, 3, 4, 5, 7, 8, 9, 10, 11\}$;
- A variável d_i representa a duração de uma atividade i;
- A variável t_{ini} representa o tempo de ínicio do modelo;
- A variável t_{fim} representa o tempo em que se findou o modelo;
- A variável r_i representa a redução da duração total da atividade i $\in \{0, 2, 3, 4, 5, 7, 8, 9, 10, 11\}$;
- A variável x_{i1} representa a redução da atividade i caso a atividade i não tenha ultrapassado a redução máxima de tempo de c_1, com i $\in \{0, 2, 3, 4, 5, 7, 8, 9, 10, 11\}$;
- A variável x_{i2} representa a redução da atividade i caso a atividade i tenha ultrapassado a redução máxima de tempo de c_1, com i $\in \{0, 2, 3, 4, 5, 7, 8, 9, 10, 11\}$;
- A variável y_{i1} toma valor 1 caso x_1 não tenha ultrapassado a redução máxima de c_1 e 0 caso contrário;
- A variável y_{i2} toma valor 1 caso x_1 tenha ultrapassado a redução máxima de c_1 e 0 caso contrário;

3.2.3 Restrições

As restrições utilizadas foram divididas em grupos:

- Restrições de precedência:
  São as restrições que estabelecem que, para uma atividade i, que **suceda** uma atividade j, o seu começo será maior ou igual que a soma do começo da atividade j com a sua duração, subtraindo a redução de tempo da atividade j.
  $$t_i \geq t_j + d_j - r_j$$
- Restrições de integralidade:
  São as restrições que definem o domínio do conjunto das soluções válidas.
  $$y_{i1}, y_{i2} \in Bin$$
- Restrições associadas às funções por partes:
  São as restrições que garantem que a modelação das funções como um conjunto de funções por partes é válida.
  $$r_i = x_{i1} + x_{i2}$$
  $$0y_{i1} \leq x_{i1} \leq m_{i1}y_{i1}$$
  $$m_{i1}y_{i2} \leq x_{i2} \leq (m_{i1} + m_{i2}) \times y_{i2}$$
  $$y_{i1} + y_{i2} = 1$$
  $$i \in \{0, 2, 3, 4, 5, 8, 10, 11\}$$

  Nota:m_{i1} *representa a redução máxima do custo* c_1 *da atividade i*
  m_{i2} *representa a redução máxima do custo* c_2 *da atividade i*

- Restrições de escolha de um conjunto discreto de opções. Representam as restrições das atividades que possuem um conjunto discreto de opções para serem escolhidas o que levará à redução da sua duração.

$$y_{i1} + y_{i2} \leq 1$$

$$x_{i1} = c_{i1} \times y_{i1}$$

$$x_{i2} = c_{i2} \times y_{i2}$$

$$i \in \{7, 9\}$$

  Nota: c_{i1} representa a quantidade reduzida pela escolha da opção representada por y_{i1}. c_{i2} representa a quantidade reduzida pela escolha da opção representada por y_{i2}.

3.3 Ficheiro de input

Decidimos apresentar o ficheiro de input desta forma de modo a ficar representado de forma mais clara, em contraste com o que aconteceria caso dispuséssemos imagens do mesmo.

```
min: 200*x01 + 50*y02 + 100*x02 +
     1000*x21 + 1500*y22 + 500*x22 +
     200*x31 + 50*y32 + 100*x32 +
     800*x41 + 800*y42 + 400*x42 +
     1600*x51 + 400*y52 + 800*x52 +
     300*y71 + 1100*y72 +
     200*x81 + 50*y82 + 100*x82 +
     200*y91 + 400*y92 +
     1000*x101 + 250*y102 + 500*x102 +
     600*x111 + 300*y112 + 300*x112;

tfim <= 21;

arco_i0: t0 >= tini + 0;
arco_i7: t7 >= tini + 0;
arco_i10: t10 >= tini + 0;

arco_02: t2 >= t0 - r0 + 4;
arco_04: t4 >= t0 - r0 + 4;

arco_23: t3 >= t2 - r2 + 7;

arco_3f: tfim >= t3 - r3 + 2 ;

arco_42: t2 >= t4 - r4 + 9 ;
arco_45: t5 >= t4 - r4 + 9 ;

arco_53: t3 >= t5 - r5 + 4 ;
arco_5f: tfim >= t5 - r5 + 4 ;
```

```
arco_74: t4 >= t7 - r7 + 6 ;
arco_78: t8 >= t7 - r7 + 6 ;

arco_85: t5 >= t8 - r8 + 4 ;
arco_89: t9 >= t8 - r8 + 4 ;

arco_9f: tfim >= t9 - r9 + 2 ;

arco_108: t8 >= t10 - r10 + 8 ;
arco_1011: t11 >= t10 - r10 + 8;

arco_119: t9 >= t11 - r11 + 7 ;

r0 = x01 + x02;
y01 + y02 = 1;
x01 >= 0*y01;
x01 <= 0.5*y01;
x02 >= 0.5*y02;
x02 <= 1*y02;

r2 = x21 + x22;
y21 + y22 = 1;
x21 >= 0*y01;
x21 <= 3*y01;
x22 >= 3*y02;
x22 <= 4*y02;

r3 = x31 + x32;
y31 + y32 = 1;
x31 >= 0*y31;
x31 <= 0.5*y31;
x32 >= 0.5*y32;
x32 <= 1*y32;

r4 = x41 + x42;
y41 + y42 = 1;
x41 >= 0*y01;
x41 <= 2*y01;
x42 >= 2*y02;
x42 <= 3*y02;

r5 = x51 + x52;
y51 + y52 = 1;
x51 >= 0*y51;
x51 <= 0.5*y51;
x52 >= 0.5*y52;
```

```
x52 <= 1*y52;

y71 + y72 <= 1;
r7 = x71 + x72;
x71 = 1*y71;
x72 = 2*y72;

r8 = x81 + x82;
y81 + y82 = 1;
x81 >= 0*y81;
x81 <= 0.5*y81;
x82 >= 0.5*y82;
x82 <= 1*y82;

y91 + y92 <= 1;
r9 = x91 + x92;
x91 = 1*y91;
x92 = 2*y92;

r10 = x101 + x102;
y101 + y102 = 1;
x101 >= 0*y101;
x101 <= 0.5*y101;
x102 >= 0.5*y102;
x102 <= 1*y102;

r11 = x111 + x112;
y111 + y112 = 1;
x111 >= 0*y111;
x111 <= 1*y111;
x112 >= 1*y112;
x112 <= 2*y112;

Bin
y01, y02, y21, y22, y31, y32, y41, y42, y51, y52,
y71, y72, y81, y82, y91, y92, y101, y102, y111, y112;
```

3.4 Ficheiro de output

Aqui apresentamos o ficheiro de output gerado. É de notar que a visualização foi ordenada pelo que as restantes variáveis que ficaram por aparecer têm todas como valor 0 no campo *result*.

Variables	MILP ...	MILP ...	re...
	2400	1250	1250
tfim	21	21	21
t3	17	20	20
t5	13	16	16
t9	15	15	15
t2	13	13	13
t8	8	8	8
t11	8	8	8
t4	6	4	4
r7	0	2	2
x72	0	2	2
x32	0	1	1
r3	0	1	1
y32	0	1	1
y72	0	1	1
y01	0	1	1
y21	1	1	1
y41	1	1	1
y51	1	1	1
y81	1	1	1
y101	1	1	1
y111	1	1	1
x01	0	0	0
y02	1	0	0
x02	0.5	0	0
x21	0	0	0
y22	0	0	0
x22	3	0	0
x31	0	0	0
x41	0	0	0
y42	0	0	0
x42	2	0	0
x51	0	0	0
y52	0	0	0
x52	0	0	0
y71	0	0	0

Figura 11: Ficheiro de output

3.5 Interpretação de resultados

Através das variáveis t_i associadas a este ficheiro de output, também conseguimos construir o **diagrama de Gantt** com as novas durações de atividades.



Figura 12: Diagrama de Gantt

Verificamos também que o custo para a redução do tempo de execução de atividades em 3 unidades de tempo é 1250 U.M.. Podemos confirmar este valor facilmente pois ao observarmos as variaveis r_i verificamos que apenas r_3 e r_7 têm valores **não nulos**. Visto que ambas pertencem ao caminho crítico desta função e que são as que representam menor custo para reduzir a duração das suas atividades, podemos verificar a veracidade do resultado.

4 Conclusão

Finda-se assim este relatório. Sabemos e apreciamos os conhecimentos que nos foram leccionados ao longo desta sequência de projetos práticos e pela disciplina de Investigação Operacional . A nivel de dificuldades podemos dizer que, felizmente não houve qualquer questão que nos tenha ocupado demasiado tempo na elaboração deste documento.